

# تطبيقات الحاسب (EMP 103)

المحاضرة الثامنة

## مقدمة الي لغة البيسك المرئي

المستوى 100 – هندسة التشييد- فصل الصيف 2018

د/ أحمد عامر شاهين

قسم هندسة الحاسبات و المنظومات

كلية الهندسة – جامعة الزقازيق

Email: [aashahin@zu.edu.eg](mailto:aashahin@zu.edu.eg)

Course Webpage:

<http://www.aashahine.faculty.zu.edu.eg/Pages/Contents.aspx?CID=30190>

- التعامل مع الكيانات الهامة.
- بناء شريط القوائم وشريط الأدوات.
- بناء صناديق الحوار والرسائل.
- المتغيرات
- السلاسل الحرفية
- اتخاذ القرار والحلقات التكرارية

• التعامل مع الملفات Files:

■ الملفات التتابعية **Sequential** والعشوائية **Random**

■ القراءة من ملف (والتأكد من وجود ملف)

■ الكتابة داخل ملف

■ الدوال الخاصة بالتعامل مع الملفات

# مخطط العرض

### الملفات :Files:

- المقصود بالملفات هي ما يتم تخزين البيانات فيها على شكل شفرة ASCII Code
- الاستفادة منها يكون في تخزين بيانات ثم استرجاعها لاحقا
- مثال:**
  - عمل ملف يحتوي على بيانات موظفين بحيث يمكن اضافة بيانات موظف جديد أو تحديث بيانات موجودة وهكذا...
  - عمل ملف يحتوي على بيانات موظفين بحيث يمكن اضافة بيانات موظف جديد أو تحديث بيانات موجودة وهكذا...

### الملفات التتابعية Sequential Files : القراءة من ملف

**فتح الملف للقراءة:**

Open filename For Input As # nFile

حيث تمثل filename الاسم المروري للملف بينما nFile تمثل رقما صحيحا (1,2,...) يرمز لللف بعد ذلك في البرنامج.

مثال: `Open "D:\Work\Students.txt" For Input As #1`

علما بأن الملف Students.txt يحتوي على البيانات التالية:

File	Edit	Format	View	Help
Amr Mohamed	90	65	75	85 70
Basem A11	67	80	72	55 82
Samy Ahmed	70	81	69	71 75

### الملفات التتابعية Sequential Files : الكتابة في ملف

**فتح الملف للكتابة:**

- وضع الاضافة:** Open filename For Append As # nFile  
أي أن المطلوب هو اضافة بيانات جديدة الى الملف الى جانب محتوياته القديمة.
- وضع الاخراج:** Open filename For Output As # nFile  
وهنا يتم حذف جميع محتويات الملف أولا ثم الكتابة فيه.
- وفى الحالتين** اذا لم يوجد ملف بهذا الاسم المذكور في المسار (filename) يتم إنشاء ملف جديد تلقائيا.

### الملفات التتابعية Sequential Files : ملاحظات عامة

**لاحظ أنه في الملفات التتابعية:**

- لا يمكن القفز أو الوصول الى سطر معين داخل الملف التتابعي ولكن يمكن فقط قراءة الأسطر واحدا تلو الآخر
- لا يمكن القراءة والكتابة في نفس الملف في وقت واحد
- هذا النوع من الملفات غير مناسب لتخزين كميات كبيرة من البيانات لأن أي قيمة يتم تخزينها على كوتها حروف (أو الحرف الواحد يحتاج Byte)

### مثال 1

صمم برنامجا يقوم بعمل جرد في مخزن حيث يتم استقبال بيانات 100 سلعة متباعدة في الاسم، العدد المتاح، سعر الوحدة. رتب البرنامج بحيث يتم تخزين هذه البيانات في ملف باسم Store ثم قم بحساب وحفظ البيانات التالية في ملف آخر باسم Available:

- اجمالي عدد السلع الموجودة مسوقا بعبارة
- بيانات السلعة ذات أعلى سعر للوحدة مسبوقة بعبارة
- Item with maximum price:
- بيانات السلعة ذات أقل عدد متاح مسبوقة بعبارة
- Item with minimum count:

**Run**

### الملفات العشوائية Random Files:

- لا بد من إنشاء (الاعلان عن) تركيبة لنوع السجلات (البيانات) التي يراد تخزينها في الملف. هذه التركيبة من نوعية ال user-defined variables (المحاضرة السابع).
- أي أن كل البيانات المخزنة لها نفس التركيبة (نفس عدد ونوعية المتغيرات).
- لذلك امكانية البحث والوصول الى سجل Record معين أفضل بكثير من الملفات التتابعية.

**مثال لتركيبة سجل**

```

Private Type Items
    sName As String
    nCount As Integer
    fPrice As Single
End Type
    
```

### مثال 2

صمم برنامجا يقوم بعمل جرد في مخزن حيث يتم استقبال بيانات 100 سلعة متباعدة في الاسم، العدد المتاح، سعر الوحدة، رتب البرنامج بحيث يتم استقبال بيانات السلع من خلال صناديق تصيرون ثم يتم ضم أسماء السلع الى صندوق موحد، بعد انتهاء ادخال بيانات كل السلع اذ اختار المستخدم أحد أسماء السلع من الصندوق الموحد يتم عرض بياناتها على نفس صناديق التصيرون المستخدمة للإدخال.

**Run**

### بعض الدوال الخاصة بالتعامل مع الملفات:

- نسخ الملفات: FileCopy source, destination
- حذف ملف: Kill filename
- اعادة تسمية ملف: Name oldname As newname
- توليد مجلد جديد: MKDir foldername

- المقصود بالملفات هي ما يتم تخزين البيانات فيها على شكل شفرة ASCII Code
- الاستفادة منها يكون في تخزين بيانات ثم استرجاعها لاحقا
- مثال:

- عمل ملف يحتوي على بيانات موظفين بحيث يمكن اضافة بيانات موظف جديد أو تحديث بيانات موجودة و هكذا...
- عمل ملف يحتوي على بيانات موظفين بحيث يمكن اضافة بيانات موظف جديد أو تحديث بيانات موجودة و هكذا...

- تنقسم الملفات إلى نوعين ملفات تتابعية Sequential Files و ملفات عشوائية Random Files
- الملفات التتابعية هي التي تتم كتابة البيانات فيها على شكل سطور
- القراءة منها تكون سطرًا سطرًا أي لا يمكن الوصول لسطر معين بدون قراءة كل الأسطر التي تسبقه
- الملفات العشوائية هي التي تكتب فيها البيانات على شكل سجلات
- يمكن قراءة أي سجل من أي مكان في الملف بمجرد ذكر رقمه لذلك سميت عشوائية
- غالبًا ما تأخذ الملفات التتابعية الامتداد (.txt) بينما الملفات العشوائية تأخذ الامتداد (.dat)


# الملفات التتابعية Sequential Files : القراءة من ملف

## فتح الملف للقراءة:

Open filename For Input As # nFile

- حيث تمثل filename الاسم المروري للملف بينما nFile تمثل رقما صحيحا (1,2,...) يرمز للملف بعد ذلك في البرنامج.

■ مثال: `Open "D:\Work\Students.txt" For Input As #1`



File	Edit	Format	View	Help			
Amr Mohamed			90	65	75	85	70
Basem Ali			67	80	72	55	82
Samy Ahmed			70	81	69	71	75

- علما بأن الملف Students.txt

يحتوى على البيانات التالية:

# الملفات التتابعية Sequential Files : القراءة من ملف

## قراءة سطر:

Line Input # nFile, stringName

## قراءة جزء من سطر على شكل مجموعة متغيرات:

Input # nFile, var1,var2,...,varn

## قراءة عدد من الأحرف من السطر:

stringName = Input (nChar, nFile)

nFile	الرقم الذي يرمز للملف
stringName	متغير سلسلة حرفية
var1,var2,...	أسماء متغيرات
nChar	عدد الأحرف

حيث أن:



# الملفات التتابعية Sequential Files : القراءة من ملف

مثال:


```
Open "D:\Work\Students.txt" For Input As #1
Line Input #1, str1
Print str1
```

Form1  
Amr Mohamed 90 65 75 85 70

```
Open "D:\Work\Students.txt" For Input As #1
str1 = Input(7, #1)
Print str1
```

Form1  
Amr Moh


■ لاحظ مسار الملف:



# الملفات التتابعية Sequential Files : القراءة من ملف

لكي نستخدم الدالة **Input** في قراءة جزء من سطر على شكل مجموعة متغيرات لابد

أن تكون هذه القيم مخزنة في الملف يفصل بينها فواصل:



File	Edit	Format	View	Help
Amr Mohamed,		90,	65,	75, 85, 70
Basem Ali,		67,	80,	72, 55, 82
Samy Ahmed,		70,	81,	69, 71, 75

```
Open "D:\Work\Students.txt" For Input As #1
Input #1, str1, n1, n2
Print str1
Print n1
Print n2
```



Form1

Amr Mohamed


90

65

# الملفات التتابعية Sequential Files : القراءة من ملف


لاحظ أن:

كل ايعاز جديد يحتوي علي **Input** أو **Line Input** ينتقل الى سطر جديد في الملف:



File	Edit	Format	View	Help
Amr Mohamed,		90,	65,	75, 85, 70
Basem Ali,		67,	80,	72, 55, 82
Samy Ahmed,		70,	81,	69, 71, 75

```
Open "D:\Work\Students.txt" For Input As #1
Line Input #1, str1
Print str1
str2 = Input(10, #1)
Print str2
```



Form1

Amr Mohamed, 90, 65, 75, 85, 70  
Basem Ali,


# الملفات التتابعية Sequential Files : القراءة من ملف

لقراءة جميع محتويات الملف نستخدم حلقة **While** مع السؤال على نهاية الملف

```
Open "D:\Work\Students.txt" For Input As #1
Do While (Not EOF(1))
    Line Input #1, str1
    Print str1
Loop
```

باستخدام الدالة **EOF**

مثال



The screenshot shows a Windows application window titled 'Form1'. The window contains a list of student names and their scores. The data is as follows:

Amr Mohamed,	90,	65,	75,	85,	70
Basem Ali,	67,	80,	72,	55,	82
Samy Ahmed,	70,	81,	69,	71,	75


# الملفات التتابعية Sequential Files : التأكد من وجود الملف

- عند فتح ملف للقراءة يفضل التأكد أولا من وجود الملف و كذلك لابد من اغلاق

```
Dim sfilepath As String, str1 As String
sfilepath = Dir$("D:\Work\Students.txt")
If (sfilepath = " ") Then
    MsgBox "File not found"
Else
    Open "D:\Work\Students.txt" For Input As #1
    Line Input #1, str1
    Print str1
End If
Close #1
```

الملف عند الانتهاء.

**مثال:**



- الدالة **Dir\$()** تبحث عن الملف و اذا كان غير موجودا ترجع سلسلة خالية.

- الدالة **Close** تغلق الملف.

# الملفات التتابعية Sequential Files : الكتابة في ملف

## فتح الملف للكتابة:

Open filename For **Append** As # nFile

■ وضع الاضافة:

أي أن المطلوب هو اضافة بيانات جديدة الى الملف الى جانب محتوياته القديمة.

Open filename For **Output** As # nFile

■ وضع الاخراج:

وهنا يتم حذف جميع محتويات الملف أولاً ثم الكتابة فيه.

■ وفي الحالتين اذا لم يوجد ملف بهذا الاسم المذكور في المسار (**filename**) يتم

إنشاء ملف جديد تلقائياً.

# الملفات التتابعية Sequential Files : الكتابة في ملف

نستخدم أي من الايعازين التاليين للكتابة في الملف:

```
Print # nFile, var1,var2,...,varn
```

```
Write # nFile, var1,var2,...,varn
```


حيث أن:

nFile	الرقم الذي يرمز للملف
var1,var2,...	أسماء متغيرات

# الملفات التتابعية Sequential Files : الكتابة في ملف

مثال:

```
Open "D:\Work\Test.txt" For Output As #1
Print #1, "aaa", "bbb"
Print #1, "ccc", "ddd";
Print #1, "xxx", "eee"
Write #1, "a1", "b1"
Write #1, "c1", "d1",
Write #1, "mmm", "zzz"
Close #1
```



```
Test - Notepad
File Edit Format View Help
aaa          bbb
ccc          dddxxx      eee
"a1", "b1"
"c1", "d1", "mmm", "zzz"
```



# الملفات التتابعية Sequential Files : الكتابة في ملف

من المثال السابق، لاحظ أن:

- علامات الاقتباس و الفواصل **لا تحسب** ضمن السلسلة الحرفية بالنسبة لايعاز **Print** بينما **تظهر** عند الكتابة باستخدام ايعاز **Write**.
- اذا **لم ينته** أي من ايعازي **Print** أو **Write** **بفصلة أو فصلة منقوطة** فانه يتم الانتقال الى **سطر جديد** في الملف.
- لابد من غلق الملف بعد الانتهاء من عملية الكتابة.

# الملفات التتابعية Sequential Files : ملاحظات عامة

لاحظ أنه في الملفات التتابعية:

- لا يمكن القفز أو الوصول الى سطر معين داخل الملف التتابعي ولكن يمكن فقط قراءة الأسطر واحدا تلو الآخر
- لا يمكن القراءة و الكتابة في نفس الملف في وقت واحد
- هذا النوع من الملفات غير مناسب لتخزين كميات كبيرة من البيانات لأن أي قيمة يتم تخزينها على كونها حروف (والحرف الواحد يحتاج 1Byte)

صمم برنامجا يقوم بعمل جرد في مخزن حيث يتم استقبال بيانات 100 سلعة متمثلة في الاسم, العدد المتاح, سعر الوحدة. رتب البرنامج بحيث يتم تخزين هذه البيانات في ملف باسم **Store** ثم قم بحساب وحفظ البيانات التالية في ملف آخر باسم **Avialable**:

- اجمالي عدد السلع الموجودة مسبقا بعبارة

"Items total count ="

- بيانات السلعة ذات أعلى سعر للوحدة مسبقه بعبارة


"Item with maximum price:"

- بيانات السلعة ذات أقل عدد متاح مسبقه بعبارة

"Item with minimum count:"

Run

```
Private Sub ComdInput_Click()  
Dim sName As String  
Dim nCount As Integer  
Dim fPrice As Single  
'receive items data  
Open "D:\Work\Store.txt" For Output As #1  
For i = 1 To 5  
    sName = InputBox("Enter the name of item # " & i)  
    nCount = Val(InputBox("Enter the count of item # " & i))  
    fPrice = Val(InputBox("Enter the price of item # " & i))  
    Write #1, sName, nCount, fPrice  
Next i  
Close #1  
End Sub
```




افتراض ادخال بيانات 5 سلع بدلا من 100


```
Private Sub CmdOutput_Click()  
Dim sName As String  
Dim nCount As Integer, nMinCount As Integer  
Dim fPrice As Single, fMaxPrice As Single  
Dim nTotCount As Integer  
'to determine total count, minimum count and maximum price  
Open "D:\Work\Store.txt" For Input As #1  
Open "D:\Work\Available.txt" For Output As #2  
Input #1, sName, nCount, fPrice  
nMinCount = nCount  
fMaxPrice = fPrice  
Do While Not EOF(1)  
    Input #1, sName, nCount, fPrice  
    nTotCount = nTotCount + nCount  
    If nCount < nMinCount Then  
        nMinCount = nCount  
    End If  
    If fPrice > fMaxPrice Then  
        fMaxPrice = fPrice  
    End If  
Loop  
Close #1  
Print #2, "Items Total Count:", nTotCount
```

```
'to get data of required items
Open "D:\Work\Store.txt" For Input As #1
Do While Not EOF(1)
    Input #1, sName, nCount, fPrice
    If nCount = nMinCount Then
        Print #2, "Item with minimum count:",
        Print #2, sName, nCount, fPrice
    End If
    If fPrice = fMaxPrice Then
        Print #2, "Item with maximum price:",
        Print #2, sName, nCount, fPrice
    End If
Loop
Close #1
Close #2
End Sub
```

# حل مثال 1: (وضع التنفيذ)



```
Store - Notepad
File Edit Format View Help
"aaa",7,12.99
"bbb",11,4.65
"ccc",10,15.9
"ddd",14,20.65
"zzz",13,17.55
```



```
Available - Notepad
File Edit Format View Help
Items Total Count:          48
Item with minimum count:   aaa          7          12.99
Item with maximum price:   ddd         14          20.65
```

- كيف يمكن عمل رؤوس قوائم في الملف Available؟ ليكون شكل الملف كالتالي:

```

Available - Notepad
File Edit Format View Help
Items Total Count: 48
Item with minimum count:
Name Count Price
aaa 7 12.99
Item with maximum price:
Name Count Price
ddd 14 20.65
    
```

- اذا كان هناك أكثر من سلعة لها نفس العدد المتاح و كذلك أكثر من سلعة لها نفس ثمن الوحدة اذن كيف نعدل حل مثال 1 لكي نتمكن من عرض جميع السلع ذات أقل عدد متاح و كذلك ذات أعلى سعر في الملف Available؟



## الملفات العشوائية Random Files:

- لا بد من إنشاء (الاعلان عن) تركيبة لنوع السجلات (البيانات) التي يراد تخزينها في الملف. هذه التركيبة من نوعية ال **user-defined variables** (المحاضرة السابعة).
- أى أن كل البيانات المخزنة لها نفس التركيبة (نفس عدد ونوعية المتغيرات).
- لذلك امكانية البحث والوصول الى سجل Record معين أفضل بكثير من الملفات

التتابعية.

```
(General) (Declarations)

Private Type Items
    sName As String
    nCount As Integer
    fPrice As Single
End Type
```

مثال لتركيبة سجل

# الملفات العشوائية Random Files:

Open filename For Random As # nFile Len= Len (record)

■ فتح ملف:

حيث أن **record** تمثل اسم متغير من النوعية المعلن عنها كتركيبية لسجلات الملف.

■ اضافة سجل الى الملف:

Put #nFile [,record number], variablename

■ استرجاع (قراءة) سجل من الملف:

Get #nFile [,record number], variablename

وتمثل **variablename** اسم متغير السجل (نفس الاسم الذي يكتب ك **record**).

ولاحظ أن **record number** حقل اختياري ويمكن تجاهله اذا لم تستدعي الحاجة.

## الملفات العشوائية Random Files:

■ **مثال:** ادخال بيانات 3 سلع ثم كتابة بيانات السلعة الأولى على النموذج:


```
(General) (Declarations)

Private Type Items
    sName As String * 10
    nCount As Integer
    fPrice As Single
End Type

Private Sub Command1_Click()
    Dim Itm1 As Items
    Dim Itm2 As Items
    Open "D:\Work\StoryInfo.dat" For Random As #1 Len = Len(Itm1)
    For i = 1 To 3
        Itm1.sName = InputBox("Enter name of item #" & i)
        Itm1.nCount = Val(InputBox("Enter count of item #" & i))
        Itm1.fPrice = Val(InputBox("Enter price of item #" & i))
        Put #1, , Itm1
    Next i
    Get #1, 1, Itm2
    Print Itm2.sName, Itm2.nCount, Itm2.fPrice
    Close #1
End Sub
```

# الملفات العشوائية Random Files:

تابع المثال:



لاحظ أنه:

- لابد من تحديد طول السلسلة الحرفية اذا وجد ضمن تركيبة السجل والا سيحدث خطأ أثناء تخزين بيانات في الملف.
- ايعاز **Get #1,1,itm2** يقوم بنسخ محتويات السجل الأول من الملف رقم 1 في المتغير itm2.

صمم برنامجا يقوم بعمل جرد في مخزن حيث يتم استقبال بيانات 100 سلعة متمثلة في الاسم, العدد المتاح, سعر الوحدة. رتب البرنامج بحيث يتم استقبال بيانات السلع من خلال صناديق نصوص ثم يتم ضم أسماء السلع الى صندوق موحد. بعد انتهاء ادخال بيانات كل السلع اذا اختار المستخدم أحد أسماء السلع من الصندوق الموحد يتم عرض بياناتها على نفس صناديق النصوص المستخدمة للادخال.

Run

## حل مثال 2: (تصميم الواجهة)

The image shows a Windows application window titled "Form1". The window contains a form with the following elements:

- Name:** A text input field.
- Count:** A text input field.
- Price:** A text input field.
- Items:** A dropdown menu.
- Start:** A button.
- Add Item:** A button.
- Show Item:** A button.
- Exit:** A red button.

## حل مثال 2: كود البرنامج

```
Private Type Items
    sName As String * 10
    nCount As Integer
    fPrice As Single
End Type
```

```
Private Sub CmdStart_Click()
Dim Itm1 As Items
Open "D:\Work\Temp.dat" For Random As #1 Len = Len(Itm1)
TxtName.SetFocus
End Sub
```

```
Private Sub CmdAdd_Click()
Dim Itm1 As Items
Itm1.sName = TxtName.Text
Itm1.nCount = Val(TxtCount.Text)
Itm1.fPrice = Val(TxtPrice.Text)
Put #1, , Itm1
Combo1.AddItem Itm1.sName
TxtName.Text = " "
TxtCount.Text = " "
TxtPrice.Text = " "
End Sub
```

## حل مثال 2: كود البرنامج

```
Private Sub CmdShow_Click()  
Dim Itm2 As Items  
Dim nRec As Integer  
nRec = Combo1.ListIndex + 1  
Get #1, nRec, Itm2  
TxtName.Text = Itm2.sName  
TxtCount.Text = Itm2.nCount  
TxtPrice.Text = Itm2.fPrice  
End Sub
```

```
Private Sub CmdExit_Click()  
Close #1  
End  
End Sub
```



## حل مثال 2: وضع التنفيذ

Form1

Name:

Count:

Price:

Items:

- aaa
- bbb
- ccc
- xxx
- zzz

Start Add Item Show Item Exit

بعد ادخال 5 سلع

Form1

Name:

Count:

Price:

Items:

- aaa
- bbb
- ccc
- xxx
- zzz

Start Add Item Show Item Exit

عرض بيانات السلعة  
المختارة

## الملفات العشوائية Random Files:

### ملاحظات:

- رقم السجل **record** داخل الملف العشوائي يبدأ من **1** ولكن دليل العناصر داخل الصندوق الموحد يبدأ من **0**, وتمت مراعاة ذلك في الايعاز: **nRec = Combo1.ListIndex + 1**
- يفضل أيضا السؤال عن وجود ملف بنفس الاسم قبل فتحه باستخدام الداله **DIR\$()**
- يمكن استخدام **EOF()** لتحديد اذا ما تم الوصول لأخر سجل في الملف أم لا
- في المثال السابق تمكنا من القراءة والكتابة في الملف العشوائي في أن واحد
- هناك ايعاز **Seek filename, recordnumber** والذي يضبط مؤشر الملف على السجل رقم **recordnumber** وبعد ذلك يمكن الدخول اليه باستخدام **Get, Put**.

- عدل حل المثال السابق بحيث يمكن اظهار عدد العناصر (السجلات) الموجودة في الملف Temp اذا أراد المستخدم ذلك
- (تعامل مع الملف مباشرة بدلا من عدّ عناصر الصندوق الموحد)
- كيف يمكن اضافة زر Remove Item بحيث اذا اختار المستخدم أحد العناصر من الصندوق الموحد ثم نقر على هذا الزر يتم ازالته من الملف ومن الصندوق أيضا؟
- (صمم طريقه بنفسك لعمل هذه الوظيفة لأنه لا يوجد ايعاز مباشر يقوم بذلك)

## بعض الدوال الخاصة بالتعامل مع الملفات:

**FileCopy** source, destination

■ نسخ الملفات:

**Kill** filename

■ حذف ملف:

**Name** oldname As newname

■ إعادة تسمية ملف:

**MKDir** foldername

■ توليد مجلد جديد:

التعامل مع الملفات سواء كانت (عشوائيه / تتابعيه)